

McXtrace: A tutorial example.

To introduce the reader to McXtrace, it is perhaps most instructive to show a very simple simulation example, which contains a point source with a constant wavelength spectrum, a platinum coated cylindrical mirror and a 2D area detector. This example is intended to serve as an entry point into the McXtrace language. In order to simulate anything at all with McXtrace, the user is required to build a structured description of the beamline describing which devices are located where. This description is contained in a file — the *instrument file*.

In the following we will show snippets of a sample instrument file. The full instrument file is supplied below as listing 1. Schematically, the experimental setup is shown in figure 1. The figure also indicates the default mode of positioning elements in McXtrace, where subsequent beamline devices are placed and oriented relative to previously defined devices. By convention, McXtrace operates with a right handed coordinate system whose default is to have z -axis point along the optical axis and the y -axis antiparallel to gravity. Any object may be re-oriented which also re-orientes any subsequent objects relative to it.

First we must have a source of photons. For sake of simplicity, we have chosen to use a flat spectrum point source. In a McXtrace instrument file this may be written as:

```
COMPONENT source = Source_pt(
  focus_xw = 0.01, focus_yh = 0.01, dist=1,
  lambda0 = 1.54, dlambda = 0.1)
AT (0, 0, 0) RELATIVE Origin
```

The above lines model a point source with a constant wavelength spectrum centred at $\lambda = 1.54 \text{ \AA}$, with a half width at half max of 0.1 \AA . The parameters *focus_xw*, *focus_yh* and *dist* (in m) are parts of the directional sampling scheme which may greatly speed up simulations by simulating only photons which have a chance of reaching the next beamline object.

The second object is a Pt-coated mirror:

```
COMPONENT mirror = Mirror_curved(
  coating = "Pt.txt", radius = 5, length = 0.2, width = 0.1)
AT (0, 0, 1) RELATIVE source
ROTATED (0, theta, 0) RELATIVE source
```

The rotation by *theta* degrees around the y -axis means that any subsequent object placed relative to the mirror is also relocated accordingly. The coating of the mirror may be chosen freely among the elements in the periodic table. To model more complex coatings additional data files may be supplied,.

Lastly, the third object is a $5 \times 5\text{mm}^2$ area detector with 101×101 pixels, placed 20 cm downstream from the mirror, defined by:

```
COMPONENT areadetector = PSD_monitor(  
  xwidth=0.005, yheight=0.005, ny=101, nx=101, filename="areadet.dat")  
AT (0,0,0.2) RELATIVE mirror
```

The detector is positioned with respect to the mirror which is rotated by θ whereas the scattering angle is 2θ . Thus the reflected beam, or part of it, might in fact miss the area detector. This problem may easily be fixed by adding an auxiliary coordinate system at the mirror's position, which is rotated by 2θ with respect to the source. In McXtrace this could be done using a model optical bench, a so called Arm, as follows:

```
COMPONENT arm = Arm()  
AT (0,0,0) RELATIVE mirror  
ROTATED (0, 2*theta, 0) RELATIVE source
```

in which case the detector should be placed relative to the Arm as: `AT (0,0,0.2) RELATIVE arm`. Using Arms makes it quite easy to model a complex setup like a 4-circle diffractometer.

Figure 2 shows the beam footprint of the reflected beam on the detector with peak intensity normalised to 1. To simulate a real world experiment it is necessary to calibrate the emitted flux of the source model to the actual conditions — otherwise we may normalise the intensity without loss of information. Figure 3 shows the observed wavelength spectrum resulting from the Pt-coating as recorded by a wavelength sensitive detector. In addition, figure 3 also shows the response from Au and Pd coatings. Note that neither these results nor the code listing 1 includes the 2θ correction arm above.

1. Code Listing

Listing 1: Complete listing of the example instrument described above

```

1  DEFINE INSTRUMENT Simple(theta=5,R=5)

    /* The DECLARE section allows us to declare variables*/
    /* functions in C syntax.*/
    DECLARE
6   %{
        double theta;

    %}

    /* The INITIALIZE section is executed once when */
11  /* the simulation starts.*/
    INITIALIZE
    %{
        if (0.2*tan(theta*DEG2RAD)>0.05){
            printf("Warning:_beam_center_will_miss_detector\n");
16    }
    %}

    TRACE

21  COMPONENT source = Source_pt(
        focus_xw = 0.01, focus_yh = 0.01,
        dist = 1, lambda0 = 1.54,
        dlambd = 0.1)
    AT (0, 0, 0) RELATIVE ABSOLUTE

26  COMPONENT mirror = Mirror_curved(
        coating = "Pt.txt", radius = R,
        length = 0.2, width = 0.1)
    AT (0, 0, 1) RELATIVE source
31  ROTATED (0, theta, 0) RELATIVE source

    COMPONENT areadetector = PSD_monitor(
        xwidth=0.05, yheight=0.05,
        ny=101, nx=101, filename="areadet.dat")
36  AT (0,0,0.2) RELATIVE mirror

END

```




Fig. 1. Sketch of a simple instrument. The related full code is given as listing 1. A mirror is placed downstream from a point source. In line with the mirror is an area detector. As θ becomes larger X-rays will miss the detector, as indicated by the upper dashed line. Also indicated in the figure are the coordinate systems of the three components in the simulation.




Fig. 2. Intensity distribution at the detector for the example beamline of section . As expected the intensity distribution is offset toward positive x since the glancing angle is $\theta = 5^\circ$ but the reflection angle is 2θ . The angle is chosen such that all numerical photons are caught by the detector.




Fig. 3. Intensity vs. wavelength for the example instrument of section , to show the wavelength response of the Pt-coating (red) of the mirror in the example. For sake of completeness we also show corresponding reflectivity curves for Pd- (blue) and Au-coatings (magenta). The glancing angle θ is set to 5° , to ensure that all rays are captured by the detector. Note that at longer wavelengths total reflection occurs even for this comparatively large glancing angle. The local maxima occurring at $\lambda \approx 5.8, 3.9$, and 5.6 \AA for Pt, Pd, and Au respectively may be related to edge energies of the elements.